# Maintainer: Sarah Ottinger <schalaalexiazeal@gmail.com>

_realname=aioresponses
pkgbase=mingw-w64-python-${_realname}
pkgname=("${MINGW_PACKAGE_PREFIX}-python-${_realname}")
pkgver=0.7.3
pkgrel=1
pkgdesc='A helper to mock/fake web requests in python aiohttp package. (mingw-w64)'
arch=('any')
mingw_arch=('mingw32' 'mingw64' 'ucrt64' 'clang64' 'clang32')
url="https://github.com/pnuckowski/aioresponses"
license=('MIT')
depends=("${MINGW_PACKAGE_PREFIX}-python-aiohttp")
checkdepends=("${MINGW_PACKAGE_PREFIX}-python-asynctest"
              "${MINGW_PACKAGE_PREFIX}-python-ddt"
              "${MINGW_PACKAGE_PREFIX}-python-pytest-runner")
makedepends=("${MINGW_PACKAGE_PREFIX}-python-pbr")
source=("${_realname}-${pkgver}.tar.gz::https://github.com/pnuckowski/aioresponses/archive/${pkgver}.tar.gz")
sha256sums=('c4a0bd1fd38680e36b7d2344ed4279062a8178c64bd66b68ffa2f8562751bf27')

export PBR_VERSION=$pkgver

prepare() {
  rm -rf "python-build-${MSYSTEM}" | true
  cp -r "${_realname}-$pkgver" "python-build-${MSYSTEM}"
}

build() {
  msg "Python build for ${MSYSTEM}"
  cd "${srcdir}/python-build-${MSYSTEM}"
  ${MINGW_PREFIX}/bin/python setup.py build
}

check() {
  cd "${srcdir}/python-build-${MSYSTEM}"
  ${MINGW_PREFIX}/bin/python setup.py pytest || warning "Tests failed"
}

package() {
  cd "${srcdir}/python-build-${MSYSTEM}"
  MSYS2_ARG_CONV_EXCL="--prefix=;--install-scripts=;--install-platlib=" \
  ${MINGW_PREFIX}/bin/python setup.py install --prefix=${MINGW_PREFIX} \
    --root="${pkgdir}" --optimize=1 --skip-build
  install -Dm644 LICENSE "${pkgdir}${MINGW_PREFIX}/share/licenses/python-${_realname}/LICENSE"
}
